



**TERVEYDEN JA
HYVINVOINNIN LAITOS**
National Institute for Health and Welfare

Health 2000/2011 Surveys

Statistical Analysis using SAS and SAS-Callable SUDAAN Packages

17.6.2013

Esa Virtala

etunimi.sukunimi@thl.fi

Terveyden ja hyvinvoinnin laitos (THL)

PL 30

00271 Helsinki

Puhelin: 029 524 6000

www.thl.fi

SAS Data Set T0011

Variables in Creation Order				
#	Variable	Type	Len	Label
1	HAVTUN	Num	8	Havaintotunnus
2	VUOSI	Num	8	Tutkimusvuosi
3	OSITE	Num	8	Osite
4	RYVAS	Num	8	Ryvas
5	ALL_ANALYSIS_W	Num	8	Painokerroin
6	SUKUPUOLI	Num	8	Sukupuoli (1=M,2=N)
7	IKA	Num	8	Ikä
8	IKA_5LK	Num	8	Ikäryhmä
9	R_AA01	Num	5	Siviilisäätö
10	R_BMI	Num	8	BMI: Painoindeksi
11	R_FS_KOL	Num	8	fS-Kol, mmol/l
12	R_SYSTBP2	Num	8	RR: syst,mittaus 2
13	R_SYSTBP2_01	Num	8	Syst.vp (0/1)
14	R_SYSTBP2_123	Num	8	Syst.vp (3-luok.)

H A V T U N	V U O S I	O S I T E	R Y V A S	A L L - A N A L Y S I S - W	S U K U P U O L I	I K A	I K A - 5 L K	R - A A 0 1	R - B M I	R - F S - K O L	R - S Y S T B P 2	R - S Y S T B P 2 - 0 1	R - S Y S T B P 2 - 1 2 3
378	2011	1	162	1.24202	1	30	1	1	27.1896	6.92000	114	0	1
383	2011	1	162	1.05219	2	40	1	1	23.2920	4.34000	120	0	1
384	2000	1	162	0.99615	2	30	1	2	29.9500	4.63739	110	0	1
	2011	1	162	0.84566	2	40	1	2	26.4236
385	2000	1	162	0.99615	2	30	1	2	23.1800	5.84660	116	0	1
386	2000	1	162	0.99615	2	35	1	2	21.0500	4.54438	114	0	1
387	2000	1	162	1.14527	1	35	1	1	24.2600	4.73041	128	0	2
	2011	1	162	0.96814	1	45	2	4	24.0741
388	2000	1	162	1.14527	1	35	1	2	27.7800	6.49771	134	0	2
389	2000	1	162	1.14527	1	35	1	5	22.4800	5.84660	118	0	1
390	2000	1	162	1.02598	2	40	1	1	22.6700	6.59073	134	0	2
	2011	1	162	0.80275	2	50	2	3	21.1927
393	2000	1	162	1.15138	1	40	1	1	22.5000	5.75359	114	0	1
395	2000	1	162	1.16129	1	50	2	1	29.2500	7.61391	136	0	2
	2011	1	162	0.88987	1	60	3	1	28.6619	4.32000	122	0	2

Descriptive Statistics / Means and Standard Errors

```
LIBNAME X "C:\T0011";
```

```
TITLE1 "Descriptive Statistics / Means and Standard Errors";
```

```
TITLE2 "SAS Means";  
PROC MEANS DATA=X.T0011 N MEAN STDERR CLM MAXDEC=3;  
  CLASS SUKUPUOLI VUOSI;  
  VAR R_SYSTBP2;  
  TYPES SUKUPUOLI*VUOSI;  
RUN;
```

```
TITLE2 "SAS Means + Weight";  
PROC MEANS DATA=X.T0011 N MEAN STDERR CLM MAXDEC=3;  
  WEIGHT ALL_ANALYSIS_W;  
  CLASS SUKUPUOLI VUOSI;  
  VAR R_SYSTBP2;  
  TYPES SUKUPUOLI*VUOSI;  
RUN;
```

```
TITLE2 "SAS Surveymeans";  
PROC SURVEYMEANS DATA=X.T0011 NOBS MEAN STDERR CLM NOMCAR;  
  STRATA OSITE;  
  CLUSTER RYVAS;  
  WEIGHT ALL_ANALYSIS_W;  
  DOMAIN SUKUPUOLI*VUOSI;  
  VAR R_SYSTBP2;  
RUN;
```

NOMCAR

requests that the procedure treat missing values in the variance computation as *not missing completely at random* (NOMCAR) for Taylor series variance estimation. When you specify the NOMCAR option, PROC SURVEYMEANS computes variance estimates by analyzing the nonmissing values as a domain or subpopulation, where the entire population includes both nonmissing and missing domains.

```
TITLE2 "SUDAAN Descript";  
PROC DESCRIPT DATA=X.T0011;  
  SETENV COLWIDTH=10 DECWIDTH=3;  
  NEST OSITE RYVAS;  
  WEIGHT ALL_ANALYSIS_W;  
  CLASS SUKUPUOLI VUOSI;  
  VAR R_SYSTBP2;  
  TABLES SUKUPUOLI*VUOSI;  
  PRINT NSUM MEAN SEMEAN LOWMEAN UPMEAN / STYLE=NCHS;  
RUN;
```

NCHS: National Center for Health Statistics, or **STYLE=BOX:** Crosstabulation

SAS Means

Analysis Variable : R_SYSTBP2 RR: syst,mittaus 2							
Sukupuoli (1=M,2=N)	Tutkimusvuosi	N Obs	N	Mean	Std Error	Lower 95% CL for Mean	Upper 95% CL for Mean
1	2000	679	648	135.204	0.741	133.748	136.659
	2011	530	434	133.283	0.833	131.647	134.920
2	2000	849	804	133.652	0.864	131.956	135.347
	2011	656	546	130.452	0.808	128.865	132.039

SAS Means + Weight

Analysis Variable : R_SYSTBP2 RR: syst,mittaus 2							
Sukupuoli (1=M,2=N)	Tutkimusvuosi	N Obs	N	Mean	Std Error	Lower 95% CL for Mean	Upper 95% CL for Mean
1	2000	679	648	134.959	0.733	133.520	136.399
	2011	530	434	132.813	0.825	131.190	134.435
2	2000	849	804	132.689	0.828	131.064	134.315
	2011	656	546	130.483	0.815	128.882	132.084

SAS Surveymeans

Domain Analysis: Sukupuoli (1=M,2=N)*Tutkimusvuosi									
Sukupuoli (1=M,2=N)	Tutkimusvuosi	Variable	Label	N	Mean	Std Error of Mean	95% CL for Mean		
1	2000	R_SYSTBP2	RR: syst,mittaus 2	648	134.959168	0.843594	133.303012	136.615325	
	2011	R_SYSTBP2	RR: syst,mittaus 2	434	132.812832	0.847909	131.148205	134.477458	
2	2000	R_SYSTBP2	RR: syst,mittaus 2	804	132.689399	0.806688	131.105687	134.273110	
	2011	R_SYSTBP2	RR: syst,mittaus 2	546	130.483271	0.833036	128.847820	132.118722	

SUDAAN Descript

Sukupuoli (1=M,2=N)	Tutkimusvuosi	Sample Size	Mean	SE Mean	Lower 95% Limit Mean	Upper 95% Limit Mean
Total	Total	2432.000	132.924	0.475	131.991	133.858
	2000	1452.000	133.769	0.645	132.503	135.036
	2011	980.000	131.586	0.527	130.552	132.621
1	Total	1082.000	134.130	0.685	132.786	135.475
	2000	648.000	134.959	0.844	133.303	136.615
	2011	434.000	132.813	0.848	131.148	134.477
2	Total	1350.000	131.833	0.661	130.535	133.130
	2000	804.000	132.689	0.807	131.106	134.273
	2011	546.000	130.483	0.833	128.848	132.119

Descriptive Statistics / Frequencies and Proportions

```
TITLE1 "Descriptive Statistics / Frequencies and Proportions";
```

```
TITLE2 "SAS Freq";  
PROC FREQ DATA=X.T0011;  
  TABLES VUOSI*R_SYSTBP2_123 / NOCOL;  
RUN;
```

```
TITLE2 "SAS Freq + weight";  
PROC FREQ DATA=X.T0011;  
  WEIGHT ALL_ANALYSIS_W;  
  TABLES VUOSI*R_SYSTBP2_123 / NOCOL;  
RUN;
```

```
TITLE2 "SAS surveyfreq";  
PROC SURVEYFREQ DATA=X.T0011 NOMCAR;  
  STRATA OSITE;  
  CLUSTER RYVAS;  
  WEIGHT ALL_ANALYSIS_W;  
  TABLES VUOSI*R_SYSTBP2_123 / ROW;  
RUN;
```

```
TITLE2 "SUDAAN Crosstab";  
PROC CROSSTAB DATA=X.T0011;  
  SETENV COLWIDTH=10 DECWIDTH=3;  
  NEST OSITE RYVAS;  
  WEIGHT ALL_ANALYSIS_W;  
  CLASS VUOSI R_SYSTBP2_123;  
  TABLES VUOSI*R_SYSTBP2_123;  
  PRINT NSUM WSUM ROWPER SEROW / STYLE=NCHS;  
RUN;
```

SAS Freq

Table of VUOSI by R_SYSTBP2_123				
VUOSI(Tutkimusvuosi)	R_SYSTBP2_123(Syst.vp (3-luok.))			
Frequency Percent Row Pct	1	2	3	Total
2000	432 17.76 29.75	850 34.95 58.54	170 6.99 11.71	1452 59.70
2011	289 11.88 29.49	626 25.74 63.88	65 2.67 6.63	980 40.30
Total	721 29.65	1476 60.69	235 9.66	2432 100.00
Frequency Missing = 282				

SAS Freq + Weight

Table of VUOSI by R_SYSTBP2_123				
VUOSI(Tutkimusvuosi)	R_SYSTBP2_123(Syst.vp (3-luok.))			
Frequency Percent Row Pct	1	2	3	Total
2000	432.307 18.32 29.89	857.608 36.33 59.29	156.428 6.63 10.82	1446.34 61.28
2011	272.169 11.53 29.78	582.749 24.69 63.75	59.1344 2.51 6.47	914.053 38.72
Total	704.476 29.85	1440.36 61.02	215.562 9.13	2360.4 100.00
Frequency Missing = 271.92682872				

SAS Surveyfreq

Table of VUOSI by R_SYSTBP2_123								
VUOSI	R_SYSTBP2_123	Frequency	Weighted Frequency	Std Dev of Wgt Freq	Percent	Std Err of Percent	Row Percent	Std Err of Row Percent
2000	1	432	432.30696	22.48752	18.3150	0.8842	29.8897	1.4174
	2	850	857.60805	26.65987	36.3332	0.9561	59.2949	1.4759
	3	170	156.42796	13.75540	6.6272	0.5757	10.8154	0.9294
	Total	1452	1446	28.18844	61.2754	0.5653	100.000	
2011	1	289	272.16918	15.57902	11.5307	0.5987	29.7761	1.4449
	2	626	582.74897	20.50978	24.6886	0.6721	63.7544	1.5555
	3	65	59.13440	7.22621	2.5053	0.3017	6.4695	0.7715
	Total	980	914.05256	24.70894	38.7246	0.5653	100.000	
Total	1	721	704.47614	30.35487	29.8457	1.1176		
	2	1476	1440	38.51085	61.0219	1.1424		
	3	235	215.56236	16.25997	9.1325	0.6734		
	Total	2432	2360	46.07662	100.000			
Frequency Missing = 282								

SUDAAN Crosstab

Tutkimusvuosi	Syst.vp (3-luok.)	Sample Size	Weighted Size	Row Percent	SE Row Percent
Total	Total	2432.000	2360.396	100.000	0.000
	1	721.000	704.476	29.846	1.118
	2	1476.000	1440.357	61.022	1.142
	3	235.000	215.562	9.132	0.673
2000	0 Total	1452.000	1446.343	100.000	0.000
	1	432.000	432.307	29.890	1.417
	2	850.000	857.608	59.295	1.476
	3	170.000	156.428	10.815	0.929
2011	1 Total	980.000	914.053	100.000	0.000
	1	289.000	272.169	29.776	1.445
	2	626.000	582.749	63.754	1.555
	3	65.000	59.134	6.469	0.771

Linear Model

```
TITLE1 "Linear Model";
```

```
TITLE2 "SAS Glm + Weight";
PROC GLM DATA=X.T0011;
  WEIGHT ALL_ANALYSIS_W;
  CLASS SUKUPUOLI VUOSI IKA_5LK;
  MODEL R_SYSTBP2 = SUKUPUOLI VUOSI SUKUPUOLI*VUOSI IKA_5LK R_BMI / SS3
        SOLUTION;
QUIT;
```

```
TITLE2 "SAS Surveyreg";
PROC SURVEYREG DATA=X.T0011 NOMCAR;
  STRATA OSITE;
  CLUSTER RYVAS;
  WEIGHT ALL_ANALYSIS_W;
  CLASS SUKUPUOLI VUOSI IKA_5LK;
  MODEL R_SYSTBP2 = SUKUPUOLI VUOSI SUKUPUOLI*VUOSI IKA_5LK R_BMI / SOLUTION
        VADJUST=NONE;
RUN;
```

VADJUST=DF | NONE

Specifies whether to use degrees of freedom adjustment $(n-1)/(n-p)$ in the computation of the matrix G for the variance estimation.

If you do not specify the VADJUST= option, by default, PROC SURVEYREG uses the degrees-of-freedom adjustment that is equivalent to the VARADJ=DF option.

If you do not want to use this variance adjustment, you can specify the VADJUST=NONE option.

```
TITLE2 "SUDAAN Regress";
PROC REGRESS DATA=X.T0011;
  SETENV COLWIDTH=10 DECWIDTH=3;
  NEST OSITE RYVAS;
  WEIGHT ALL_ANALYSIS_W;
  CLASS SUKUPUOLI VUOSI IKA_5LK;
  MODEL R_SYSTBP2 = SUKUPUOLI VUOSI SUKUPUOLI*VUOSI IKA_5LK R_BMI;
  PREDMARG SUKUPUOLI VUOSI SUKUPUOLI*VUOSI IKA_5LK;
  PRINT BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

PREDMARG

The predicted marginal is defined as the average predicted response if all the observations had been in a given group or were at a specified value for a continuous variable.

For the theory and application as well as practical interpretation of conditional and predicted marginals, one may refer to Graubard and Korn (1998), Korn and Graubard (1999), Neter, Wasserman, and Kutner (1990), Lee (1981), Lane and Nelder (1982), and Chang, Gelman, and Pagano (1982).

SAS Glim + Weight

Source	DF	Type III SS	Mean Square	F Value	Pr > F
SUKUPUOLI	1	3131.0531	3131.0531	10.23	0.0014
VUOSI	1	12093.6383	12093.6383	39.49	<.0001
SUKUPUOLI*VUOSI	1	198.3798	198.3798	0.65	0.4210
IKA_5LK	4	146446.4323	36611.6081	119.56	<.0001
R_BMI	1	32834.4980	32834.4980	107.23	<.0001

Parameter	Estimate		Standard Error	t Value	Pr > t
Intercept	118.9016146	B	2.47231857	48.09	<.0001
SUKUPUOLI 1	1.7855588	B	1.16243197	1.54	0.1247
SUKUPUOLI 2	0.0000000	B	.	.	.
VUOSI 2000	4.1134266	B	1.02602639	4.01	<.0001
VUOSI 2011	0.0000000	B	.	.	.
SUKUPUOLI*VUOSI 1 2000	1.1951572	B	1.48486288	0.80	0.4210
SUKUPUOLI*VUOSI 1 2011	0.0000000	B	.	.	.
SUKUPUOLI*VUOSI 2 2000	0.0000000	B	.	.	.
SUKUPUOLI*VUOSI 2 2011	0.0000000	B	.	.	.
IKA_5LK 1	-21.6183723	B	1.30449710	-16.57	<.0001
IKA_5LK 2	-13.7822737	B	1.33116245	-10.35	<.0001
IKA_5LK 3	-6.9280087	B	1.37527764	-5.04	<.0001
IKA_5LK 4	-1.2758917	B	1.45652787	-0.88	0.3811
IKA_5LK 5	0.0000000	B	.	.	.
R_BMI	0.8125678		0.07847014	10.36	<.0001

SAS Surveyreg

Tests of Model Effects			
Effect	Num DF	F Value	Pr > F
Model	8	102.17	<.0001
Intercept	1	2433.33	<.0001
SUKUPUOLI	1	7.77	0.0055
VUOSI	1	42.01	<.0001
SUKUPUOLI*VUOSI	1	1.02	0.3138
IKA_5LK	4	128.29	<.0001
R_BMI	1	93.32	<.0001

Estimated Regression Coefficients				
Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	118.901615	2.40197655	49.50	<.0001
SUKUPUOLI 1	1.785559	1.16013046	1.54	0.1242
SUKUPUOLI 2	0.000000	0.00000000	.	.
VUOSI 2000	4.113427	0.86316278	4.77	<.0001
VUOSI 2011	0.000000	0.00000000	.	.
SUKUPUOLI*VUOSI 1 2000	1.195157	1.18574191	1.01	0.3138

Estimated Regression Coefficients				
Parameter	Estimate	Standard Error	t Value	Pr > t
SUKUPUOLI*VUOSI 1 2011	0.000000	0.00000000	.	.
SUKUPUOLI*VUOSI 2 2000	0.000000	0.00000000	.	.
SUKUPUOLI*VUOSI 2 2011	0.000000	0.00000000	.	.
IKA_5LK 1	-21.618372	1.66532293	-12.98	<.0001
IKA_5LK 2	-13.782274	1.78618522	-7.72	<.0001
IKA_5LK 3	-6.928009	1.84147998	-3.76	0.0002
IKA_5LK 4	-1.275892	1.99001651	-0.64	0.5216
IKA_5LK 5	0.000000	0.00000000	.	.
R_BMI	0.812568	0.08411401	9.66	<.0001

SUDAAN Regress

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		118.902	2.402	49.502	0.000
Sukuoli (1=M,2=N)	1	1.786	1.160	1.539	0.124
	2	0.000	0.000	.	.
Tutkimusvuosi	2000	4.113	0.863	4.766	0.000
	2011	0.000	0.000	.	.
Ikäryhmä	1	-21.618	1.665	-12.981	0.000
	2	-13.782	1.786	-7.716	0.000
	3	-6.928	1.841	-3.762	0.000
	4	-1.276	1.990	-0.641	0.522
	5	0.000	0.000	.	.
BMI: Painoindeksi		0.813	0.084	9.660	0.000
Sukuoli (1=M,2=N), Tutkimusvuosi	1, 2000	1.195	1.186	1.008	0.314
	1, 2011	0.000	0.000	.	.
	2, 2000	0.000	0.000	.	.
	2, 2011	0.000	0.000	.	.

Contrast	Degrees of Freedom	Wald F	P-value Wald F
OVERALL MODEL	9.000	11752.783	0.000
MODEL MINUS INTERCEPT	8.000	102.169	0.000
INTERCEPT	.	.	.
SUKUPUOLI	.	.	.
VUOSI	.	.	.
IKA_5LK	4.000	128.292	0.000
R_BMI	1.000	93.322	0.000
SUKUPUOLI * VUOSI	1.000	1.016	0.314

Predicted Marginal #1		Predicted Marginal	Lower 95% Limit	Upper 95% Limit
Sukupuoli (1=M,2=N)	1	134.267	132.965	135.569
	2	131.751	130.591	132.911
Tutkimusvuosi	2000	134.768	133.524	136.012
	2011	130.086	129.075	131.097
Ikäryhmä	1	122.838	121.671	124.005
	2	130.674	129.273	132.074
	3	137.528	135.656	139.401
	4	143.180	140.861	145.499
	5	144.456	141.427	147.486
Sukupuoli (1=M,2=N), Tutkimusvuosi	1, 2000	136.330	134.712	137.949
	1, 2011	131.022	129.387	132.657
	2, 2000	133.350	131.925	134.775
	2, 2011	129.236	127.827	130.646

Domain Analysis / Continuous Response

```
TITLE1 "Domain Analysis / Continuous Response";

TITLE2 "SAS surveyreg / Linear Model";
PROC SURVEYREG DATA=X.T0011 NOMCAR;
  STRATA  OSITE;
  CLUSTER RYVAS;
  WEIGHT  ALL_ANALYSIS_W;
  DOMAIN  VUOSI;
  CLASS   SUKUPUOLI;
  MODEL   R_SYSTBP2 = SUKUPUOLI SUKUPUOLI*IKA / SOLUTION VARADJ=NONE;
RUN;
```

DOMAIN

The DOMAIN statement requests analysis for subpopulations, or domains, in addition to analysis for the entire study population. The DOMAIN statement names the variables that identify domains, which are called domain variables.

It is common practice to compute statistics for domains. The formation of these domains might be unrelated to the sample design. Therefore, the sample sizes for the domains are random variables. In order to incorporate this variability into the variance estimation, you should use a DOMAIN statement.

Note that a DOMAIN statement is different from a **BY** statement. In a BY statement, you treat the sample sizes as fixed in each subpopulation, and you perform analysis within each BY group independently. You should use the DOMAIN statement on the entire data set to perform the domain analysis. Creating a new data set from a single domain and analyzing that with SURVEYREG yields inappropriate estimates of variance

```
TITLE2 "SUDAAN Regress / Linear Model";
PROC REGRESS DATA=X.T0011 ;
  SETENV COLWIDTH=10 DECWIDTH=3;
  NEST    OSITE RYVAS;
  WEIGHT  ALL_ANALYSIS_W;
  RBY     VUOSI;
  CLASS   SUKUPUOLI;
  MODEL   R_SYSTBP2 = SUKUPUOLI SUKUPUOLI*IKA;
  PREDMARG SUKUPUOLI
  PRINT   BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

RBY

Using a RBY statement in SUDAAN is not equivalent to using a subset of the data file, where the observations you wish to exclude have been deleted.

Differences (usually slight) will be evident in estimates of standard errors, and are not due to the SRS-based variance-covariance matrix, but to the difference in the robust variance-covariance matrix.

These differences arise from differences in counts within the strata (of PSUs); using RBY corresponds to the assumption that, even if there are no individuals in a PSU in the sample, there may be some in the universe, and an appropriate contribution to the estimated variance must be calculated.

TITLE2 "CAUTION: SAS Surveyreg Using BY-Group Processing";

```
PROC SORT DATA=X.T0011 OUT=TMP;  
  BY VUOSI;  
RUN;
```

```
PROC SURVEYREG DATA=tmp NOMCAR;  
  STRATA OSITE;  
  CLUSTER RYVAS;  
  WEIGHT ALL_ANALYSIS_W;  
  BY VUOSI;  
  CLASS SUKUPUOLI;  
  MODEL R_SYSTBP2 = SUKUPUOLI SUKUPUOLI*IKA / SOLUTION VARADJ=NONE;  
RUN;
```

BY

You can specify a BY statement with PROC SURVEYREG to obtain separate analyses on observations in groups defined by the BY variables.

Note that using a BY statement provides COMPLETELY SEPARATE analyses of the BY groups.

IT DOES NOT provide a statistically valid subpopulation or domain analysis, where the total number of units in the subpopulation is not known with certainty.

For more information about subpopulation analysis for sample survey data, see Cochran (1977).

SAS Surveyreg / Linear Model

Domain Regression Analysis for Variable R_SYSTBP2

Tutkimusvuosi=2000

Tests of Model Effects			
Effect	Num DF	F Value	Pr > F
Model	3	131.41	<.0001
Intercept	1	2853.84	<.0001
SUKUPUOLI	1	48.16	<.0001
IKA*SUKUPUOLI	2	191.24	<.0001

Estimated Regression Coefficients				
Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	89.6026311	2.29931659	38.97	<.0001
SUKUPUOLI 1	23.7345720	3.42006527	6.94	<.0001
SUKUPUOLI 2	0.0000000	0.00000000	.	.
IKA*SUKUPUOLI 1	0.4389652	0.05937355	7.39	<.0001
IKA*SUKUPUOLI 2	0.8330642	0.04416157	18.86	<.0001

Tutkimusvuosi=2011

Tests of Model Effects			
Effect	Num DF	F Value	Pr > F
Model	3	58.56	<.0001
Intercept	1	3169.12	<.0001
SUKUPUOLI	1	30.59	<.0001
IKA*SUKUPUOLI	2	76.80	<.0001

Estimated Regression Coefficients				
Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	97.7840346	2.71554712	36.01	<.0001
SUKUPUOLI 1	22.5840989	4.08360932	5.53	<.0001
SUKUPUOLI 2	0.0000000	0.00000000	.	.
IKA*SUKUPUOLI 1	0.2285176	0.05359577	4.26	<.0001
IKA*SUKUPUOLI 2	0.5941704	0.05220833	11.38	<.0001

SUDAAN Regress / Linear Model

Results for BY-Group: Tutkimusvuosi = 2000

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		89.603	2.299	38.969	0.000
Sukupuoli (1=M,2=N)	1	23.735	3.420	6.940	0.000
	2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	0.439	0.059	7.393	0.000
	2, 1	0.833	0.044	18.864	0.000

Contrast	Degrees of Freedom	Wald F	P-value Wald F
OVERALL MODEL	4.000	14723.221	0.000
MODEL MINUS INTERCEPT	3.000	131.411	0.000
INTERCEPT	.	.	.
SUKUPUOLI	1.000	48.161	0.000
IKA * SUKUPUOLI	2.000	191.237	0.000

Predicted Marginal #1	Predicted Marginal	Lower 95% Limit	Upper 95% Limit	
Sukupuoli (1=M,2=N)	1	135.526	133.858	137.194
	2	131.713	130.279	133.146

Results for BY-Group: Tutkimusvuosi = 2011

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		97.784	2.716	36.009	0.000
Sukupuoli (1=M,2=N)	1	22.584	4.084	5.530	0.000
	2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	0.229	0.054	4.264	0.000
	2, 1	0.594	0.052	11.381	0.000

Contrast	Degrees of Freedom	Wald F	P-value Wald F
OVERALL MODEL	4.000	17769.239	0.000
MODEL MINUS INTERCEPT	3.000	58.563	0.000
INTERCEPT	.	.	.
SUKUPUOLI	1.000	30.586	0.000
IKA * SUKUPUOLI	2.000	76.799	0.000

Predicted Marginal #1	Predicted Marginal	Lower 95% Limit	Upper 95% Limit	
Sukupuoli (1=M,2=N)	1	132.882	131.261	134.503
	2	130.321	128.787	131.856

Domain Analysis / Binary(0/1) Response

```
TITLE1 "Domain Analysis / Binary(0/1) Response";

TITLE2 "SAS Surveylogistic / Binary Logit Model";
PROC SURVEYLOGISTIC DATA=X.T0011 NOMCAR;
  STRATA  OSITE;
  CLUSTER RYVAS;
  WEIGHT  ALL_ANALYSIS_W;
  DOMAIN  VUOSI;
  CLASS   SUKUPUOLI / PARAM=GLM;
  MODEL   R_SYSTBP2_01(EVENT='1') = SUKUPUOLI SUKUPUOLI*IKA / VADJUST=NONE
          TECHNIQUE=NEWTON;
RUN;
```

TECHNIQUE=FISHER | NEWTON

Specifies the optimization technique for estimating the regression parameters.

NEWTON (or NR) is the Newton-Raphson algorithm and FISHER (or FS) is the Fisher scoring algorithm.

Both techniques yield the same estimates, but the estimated covariance matrices are slightly different except for the case where the LOGIT link is specified for binary response data.

The default is TECHNIQUE=FISHER.

```
TITLE2 "SUDAAN Rlogist / Binary Logit Model";
PROC RLOGIST DATA=X.T0011 ;
  SETENV  COLWIDTH=10 DECWIDTH=3;
  NEST    OSITE RYVAS;
  WEIGHT  ALL_ANALYSIS_W;
  RBY     VUOSI;
  CLASS   SUKUPUOLI;
  MODEL   R_SYSTBP2_01 = SUKUPUOLI SUKUPUOLI*IKA;
  TEST    WALDCHI;
  PREDMARG SUKUPUOLI
  PRINT   BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

SAS Surveylogistic / Binary Logit Model

Domain Analysis for domain Tutkimusvuosi=2000

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr > ChiSq
SUKUPUOLI	1	15.5693	<.0001
IKA*SUKUPUOLI	2	236.5157	<.0001

Analysis of Maximum Likelihood Estimates						
Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		1	-5.2226	0.3662	203.4395	<.0001
SUKUPUOLI	1	1	2.0616	0.5225	15.5693	<.0001
SUKUPUOLI	2	0	0	.	.	.
IKA*SUKUPUOLI	1	1	0.0486	0.00650	55.9839	<.0001
IKA*SUKUPUOLI	2	1	0.0819	0.00650	158.8352	<.0001

Domain Analysis for domain Tutkimusvuosi=2011

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr > ChiSq
SUKUPUOLI	1	14.5444	0.0001
IKA*SUKUPUOLI	2	82.3609	<.0001

Analysis of Maximum Likelihood Estimates						
Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		1	-4.8276	0.4553	112.4397	<.0001
SUKUPUOLI	1	1	2.5596	0.6712	14.5444	0.0001
SUKUPUOLI	2	0	0	.	.	.
IKA*SUKUPUOLI	1	1	0.0221	0.00716	9.5498	0.0020
IKA*SUKUPUOLI	2	1	0.0634	0.00791	64.3252	<.0001

SUDAAN Rlogist / Binary Logit Model

Results for BY-Group: Tutkimusvuosi = 2000

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		-5.223	0.366	-14.263	0.000
Sukupuoli (1=M,2=N)	1	2.062	0.522	3.946	0.000
	2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	0.049	0.006	7.482	0.000
	2, 1	0.082	0.006	12.603	0.000

Contrast	Degrees of Freedom	Wald ChiSq	P-value Wald ChiSq
OVERALL MODEL	4.000	346.685	0.000
MODEL MINUS INTERCEPT	3.000	240.667	0.000
INTERCEPT	.	.	.
SUKUPUOLI	1.000	15.569	0.000
IKA * SUKUPUOLI	2.000	236.516	0.000

Predicted Marginal #1		Predicted Marginal	Lower 95% Limit	Upper 95% Limit
Sukupuoli (1=M,2=N)	1	0.346	0.310	0.383
	2	0.298	0.270	0.327

Results for BY-Group: Tutkimusvuosi = 2011

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		-4.829	0.455	-10.603	0.000
Sukupuoli (1=M,2=N)	1	2.561	0.671	3.815	0.000
	2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	0.022	0.007	3.090	0.002
	2, 1	0.063	0.008	8.021	0.000

Contrast	Degrees of Freedom	Wald ChiSq	P-value Wald ChiSq
OVERALL MODEL	4.000	375.540	0.000
MODEL MINUS INTERCEPT	3.000	90.003	0.000
INTERCEPT	.	.	.
SUKUPUOLI	1.000	14.551	0.000
IKA * SUKUPUOLI	2.000	82.366	0.000

Predicted Marginal #1		Predicted Marginal	Lower 95% Limit	Upper 95% Limit
Sukupuoli (1=M,2=N)	1	0.263	0.228	0.302
	2	0.242	0.207	0.280

Domain Analysis / Categorical Response

```
TITLE1 "Domain Analysis / Categorical Response";

TITLE2 "SAS Surveylogistic / Cumulative Logit Model";
PROC SURVEYLOGISTIC DATA=X.T0011 NOMCAR;
  STRATA  OSITE;
  CLUSTER RYVAS;
  WEIGHT  ALL_ANALYSIS_W;
  DOMAIN  VUOSI;
  CLASS   SUKUPUOLI / PARAM=GLM;
  MODEL   R_SYSTBP2_123 = SUKUPUOLI SUKUPUOLI*IKA / VADJUST=NONE
          TECHNIQUE=NEWTON LINK=CLOGIT;
RUN;
```

LINK=keyword

Specifies the link function that links the response probabilities to the linear predictors. The default is LINK=LOGIT.

LOGIT

Specifies the cumulative logit function. PROC SURVEYLOGISTIC fits the binary logit model when there are two response categories and fits the cumulative logit model when there are more than two response categories. Aliases: CLOGIT, CUMLOGIT.

GLOGIT

Specifies the generalized logit function. PROC SURVEYLOGISTIC fits the generalized logit model where each nonreference category is contrasted with the reference category.

You can use the response variable option REF= to specify the reference category.

```
TITLE2 "SUDAAN Multilog / Cumulative Logit Model";
PROC MULTILOG DATA=X.T0011 ;
  SETENV COLWIDTH=10 DECWIDTH=3;
  NEST   OSITE RYVAS;
  WEIGHT ALL_ANALYSIS_W;
  RBY    VUOSI;
  CLASS  R_SYSTBP2_123 SUKUPUOLI;
  MODEL  R_SYSTBP2_123 = SUKUPUOLI SUKUPUOLI*IKA / CUMLOGIT;
  TEST   WALDCHI;
  PREDMARG SUKUPUOLI
  PRINT  BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

CUMLOGIT

Use the CUMLOGIT (cumulative logit) option for ORDINAL responses.

The CUMLOGIT link function specifies the Proportional Odds Model

GENLOGIT

Use the default GENLOGIT (generalized logit) option for NOMINAL responses.

The GENLOGIT link function specifies the Multinomial Logit Model

SAS Surveylogistic / Cumulative Logit Model

Domain Analysis for domain Tutkimusvuosi=2000

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr > ChiSq
SUKUPUOLI	1	31.1045	<.0001
IKA*SUKUPUOLI	2	242.0963	<.0001

Analysis of Maximum Likelihood Estimates						
Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept	1	1	3.2134	0.2761	135.4499	<.0001
Intercept	2	1	6.7153	0.3385	393.4956	<.0001
SUKUPUOLI	1	1	-2.1719	0.3894	31.1045	<.0001
SUKUPUOLI	2	0	0	.	.	.
IKA*SUKUPUOLI	1	1	-0.0461	0.00640	51.9046	<.0001
IKA*SUKUPUOLI	2	1	-0.0788	0.00531	219.9552	<.0001

Domain Analysis for domain Tutkimusvuosi=2011

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr > ChiSq
SUKUPUOLI	1	19.9301	<.0001
IKA*SUKUPUOLI	2	110.9704	<.0001

Analysis of Maximum Likelihood Estimates						
Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept	1	1	3.1134	0.3799	67.1497	<.0001
Intercept	2	1	7.0981	0.4639	234.0717	<.0001
SUKUPUOLI	1	1	-2.5553	0.5724	19.9301	<.0001
SUKUPUOLI	2	0	0	.	.	.
IKA*SUKUPUOLI	1	1	-0.0309	0.00834	13.7100	0.0002
IKA*SUKUPUOLI	2	1	-0.0728	0.00713	104.1090	<.0001

SUDAAN Multilog / Cumulative Logit Model

Results for BY-Group: Tutkimusvuosi = 2000

R_SYSTBP2_123 (cum- logit), Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
R_SYSTBP2_123 (cum- logit)	Intercept 1	3.213	0.276	11.638	0.000
	Intercept 2	6.715	0.339	19.837	0.000
Sukupuoli (1=M,2=N)	1	-2.172	0.389	-5.577	0.000
	2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	-0.046	0.006	-7.204	0.000
	2, 1	-0.079	0.005	-14.831	0.000

Contrast	Degrees of Freedom	Wald ChiSq	P-value Wald ChiSq
OVERALL MODEL	5.000	822.292	0.000
MODEL MINUS INTERCEPT	3.000	244.882	0.000
SUKUPUOLI	1.000	31.104	0.000
IKA * SUKUPUOLI	2.000	242.096	0.000

Syst.vp (3-luok.)	Predicted Marginal #1	Predicted Marginal	Lower 95% Limit	Upper 95% Limit
1	Sukupuoli (1=M,2=N) 1	0.237	0.205	0.271
	2	0.361	0.328	0.395
2	Sukupuoli (1=M,2=N) 1	0.648	0.613	0.681
	2	0.540	0.509	0.570
3	Sukupuoli (1=M,2=N) 1	0.116	0.093	0.143
	2	0.099	0.082	0.120

Results for BY-Group: Tutkimusvuosi = 2011

R_SYSTBP2_123 (cum- logit), Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
R_SYSTBP2_123 (cum- logit)	Intercept 1	3.114	0.380	8.195	0.000
	Intercept 2	7.099	0.464	15.299	0.000
Sukupuoli (1=M,2=N)	1	-2.555	0.572	-4.464	0.000
	2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	-0.031	0.008	-3.703	0.000
	2, 1	-0.073	0.007	-10.203	0.000

Contrast	Degrees of Freedom	Wald ChiSq	P-value Wald ChiSq
OVERALL MODEL	5.000	579.476	0.000
MODEL MINUS INTERCEPT	3.000	113.217	0.000
SUKUPUOLI	1.000	19.930	0.000
IKA * SUKUPUOLI	2.000	110.968	0.000

Syst.vp (3-luok.)	Predicted Marginal #1	Predicted Marginal	Lower 95% Limit	Upper 95% Limit
1	Sukupuoli (1=M,2=N) 1	0.253	0.215	0.296
	2	0.335	0.298	0.375
2	Sukupuoli (1=M,2=N) 1	0.687	0.649	0.723
	2	0.597	0.559	0.633
3	Sukupuoli (1=M,2=N) 1	0.060	0.043	0.082
	2	0.068	0.052	0.088

Follow-Up Study, Repeated Measures / Continuous Response

```
TITLE1 "Follow-Up Study, Repeated Measures / Continuous Response";

PROC SORT DATA=X.T0011 OUT=TMP;
  BY OSITE HAVTUN;
RUN;

TITLE2 "SAS Surveyreg / Linear Model";
PROC SURVEYREG DATA=TMP NOMCAR;
  STRATA OSITE;
  CLUSTER HAVTUN;
  WEIGHT ALL_ANALYSIS_W;
  CLASS VUOSI SUKUPUOLI;
  MODEL R_SYSTBP2 = SUKUPUOLI VUOSI*SUKUPUOLI SUKUPUOLI*IKA / SOLUTION
  VADJUST=NONE;
RUN;

TITLE2 "SUDAAN Regress / Linear Model";
PROC REGRESS DATA=TMP ;
  SETENV COLWIDTH=10 DECWIDTH=3;
  NEST OSITE HAVTUN;
  WEIGHT ALL_ANALYSIS_W;
  CLASS VUOSI SUKUPUOLI;
  MODEL R_SYSTBP2 = SUKUPUOLI VUOSI*SUKUPUOLI SUKUPUOLI*IKA;
  PREDMARG VUOSI*SUKUPUOLI;
  PRINT BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

SAS Surveyreg / Linear Model

Tests of Model Effects			
Effect	Num DF	F Value	Pr > F
Model	5	109.73	<.0001
Intercept	1	5827.04	<.0001
SUKUPUOLI	1	73.36	<.0001
VUOSI*SUKUPUOLI	2	22.95	<.0001
IKA*SUKUPUOLI	2	253.08	<.0001

Estimated Regression Coefficients				
Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	89.8273237	1.94411014	46.20	<.0001
SUKUPUOLI 1	23.7034207	2.95202040	8.03	<.0001
SUKUPUOLI 2	0.0000000	0.00000000	.	.
VUOSI*SUKUPUOLI 2000 1	3.9881058	0.97120393	4.11	<.0001
VUOSI*SUKUPUOLI 2000 2	4.6533162	0.86448335	5.38	<.0001
VUOSI*SUKUPUOLI 2011 1	0.0000000	0.00000000	.	.
VUOSI*SUKUPUOLI 2011 2	0.0000000	0.00000000	.	.
IKA*SUKUPUOLI 1	0.3540702	0.04040819	8.76	<.0001
IKA*SUKUPUOLI 2	0.7387500	0.03569765	20.69	<.0001

SUDAAN Regress / Linear Model

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		89.827	1.944	46.205	0.000
Sukupuoli (1=M,2=N)	1	23.703	2.952	8.030	0.000
	2	0.000	0.000	.	.
Tutkimusvuosi, Sukupuoli (1=M,2=N)	2000, 1	3.988	0.971	4.106	0.000
	2000, 2	4.653	0.864	5.383	0.000
	2011, 1	0.000	0.000	.	.
	2011, 2	0.000	0.000	.	.
Sukupuoli (1=M,2=N), Ikä	1, 1	0.354	0.040	8.762	0.000
	2, 1	0.739	0.036	20.695	0.000

Contrast	Degrees of Freedom	Wald F	P-value Wald F
OVERALL MODEL	6.000	18957.803	0.000
MODEL MINUS INTERCEPT	5.000	109.730	0.000
INTERCEPT	.	.	.
SUKUPUOLI	.	.	.
VUOSI * SUKUPUOLI	2.000	22.954	0.000
IKA * SUKUPUOLI	2.000	253.084	0.000

Predicted Marginal #1	Predicted Marginal	Lower 95% Limit	Upper 95% Limit	
Tutkimusvuosi, Sukupuoli (1=M,2=N)	2000, 1	135.994	134.532	137.456
	2000, 2	133.028	131.550	134.507
	2011, 1	132.006	130.422	133.590
	2011, 2	128.375	126.913	129.837

Follow-Up Study, Repeated Measures / Binary(0/1) Response

```
TITLE1 "Follow-Up Study, Repeated Measures / Binary(0/1) Response";

TITLE2 "SAS Surveylogistic / Binary Logit Model";
PROC SURVEYLOGISTIC DATA=TMP NOMCAR;
  STRATA  OSITE;
  CLUSTER HAVTUN;
  WEIGHT  ALL_ANALYSIS_W;
  CLASS  VUOSI SUKUPUOLI / PARAM=GLM;
  MODEL  R_SYSTBP2_01(EVENT='1') = SUKUPUOLI VUOSI*SUKUPUOLI SUKUPUOLI*IKA /
        VADJUST=NONE TECHNIQUE=NEWTON;
RUN;

TITLE2 "SUDAAN Rlogist / Binary Logit Model";
PROC RLOGIST DATA=TMP ;
  SETENV  COLWIDTH=10 DECWIDTH=3;
  NEST    OSITE HAVTUN;
  WEIGHT  ALL_ANALYSIS_W;
  CLASS  VUOSI SUKUPUOLI;
  MODEL  R_SYSTBP2_01 = SUKUPUOLI VUOSI*SUKUPUOLI SUKUPUOLI*IKA;
  TEST   WALDCHI;
  PREDMARG VUOSI*SUKUPUOLI;
  PRINT  BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

SAS Surveylogistic / Binary Logit Model

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr > ChiSq
SUKUPUOLI	1	31.5079	<.0001
VUOSI*SUKUPUOLI	2	45.9609	<.0001
IKA*SUKUPUOLI	2	287.7558	<.0001

Analysis of Maximum Likelihood Estimates							
Parameter			DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept			1	-5.5358	0.3171	304.7124	<.0001
SUKUPUOLI	1		1	2.3295	0.4429	27.6641	<.0001
SUKUPUOLI	2		0	0	.	.	.
VUOSI*SUKUPUOLI	2000	1	1	0.5644	0.1403	16.1927	<.0001
VUOSI*SUKUPUOLI	2000	2	1	0.7055	0.1292	29.8078	<.0001
VUOSI*SUKUPUOLI	2011	1	0	0	.	.	.
VUOSI*SUKUPUOLI	2011	2	0	0	.	.	.
IKA*SUKUPUOLI	1		1	0.0385	0.00478	65.0599	<.0001
IKA*SUKUPUOLI	2		1	0.0749	0.00503	221.9152	<.0001

SUDAAN Rlogist / Binary Logit Model

Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
Intercept		-5.536	0.317	-17.456	0.000
Sukuupuoli (1=M,2=N)	1	2.330	0.443	5.260	0.000
	2	0.000	0.000	.	.
Tutkimusvuosi, Sukuupuoli (1=M,2=N)	2000, 1	0.564	0.140	4.024	0.000
	2000, 2	0.706	0.129	5.460	0.000
	2011, 1	0.000	0.000	.	.
	2011, 2	0.000	0.000	.	.
Sukuupuoli (1=M,2=N), Ikä	1, 1	0.039	0.005	8.066	0.000
	2, 1	0.075	0.005	14.897	0.000

Contrast	Degrees of Freedom	Wald ChiSq	P-value Wald ChiSq
OVERALL MODEL	6.000	547.166	0.000
MODEL MINUS INTERCEPT	5.000	299.326	0.000
INTERCEPT	.	.	.
SUKUPUOLI	.	.	.
VUOSI * SUKUPUOLI	2.000	45.961	0.000
IKA * SUKUPUOLI	2.000	287.756	0.000

Predicted Marginal #1		Predicted Marginal	Lower 95% Limit	Upper 95% Limit
Tutkimusvuosi, Sukuupuoli (1=M,2=N)	2000, 1	0.357	0.321	0.394
	2000, 2	0.322	0.291	0.354
	2011, 1	0.247	0.209	0.289
	2011, 2	0.213	0.183	0.246

Follow-Up Study, Repeated Measures / Categorical Response

```
TITLE1 "Follow-Up Study, Repeated Measures / Categorical Response";

TITLE2 "SAS Surveylogistic / Cumulative Logit Model";
PROC SURVEYLOGISTIC DATA=TMP NOMCAR;
  STRATA  OSITE;
  CLUSTER HAVTUN;
  WEIGHT  ALL_ANALYSIS_W;
  CLASS   VUOSI SUKUPUOLI / PARAM=GLM;
  MODEL   R_SYSTBP2_123 = SUKUPUOLI VUOSI*SUKUPUOLI SUKUPUOLI*IKA /
           VADJUST=NONE TECHNIQUE=NEWTON LINK=CLOGIT;
RUN;

TITLE2 "SUDAAN Multilog / Cumulative Logit Model";
PROC MULTILOG DATA=TMP ;
  SETENV  COLWIDTH=10 DECWIDTH=3;
  NEST    OSITE HAVTUN;
  WEIGHT  ALL_ANALYSIS_W;
  CLASS   R_SYSTBP2_123 VUOSI SUKUPUOLI;
  MODEL   R_SYSTBP2_123 = SUKUPUOLI VUOSI*SUKUPUOLI SUKUPUOLI*IKA / CUMLOGIT;
  TEST    WALDCHI;
  PREDMARG VUOSI*SUKUPUOLI;
  PRINT   BETA SEBETA T_BETA P_BETA PREDMRG LOWPM UPPM / TESTS=DEFAULT;
RUN;
```

SAS Surveylogistic / Cumulative Logit Model

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr > ChiSq
SUKUPUOLI	1	44.9026	<.0001
VUOSI*SUKUPUOLI	2	22.2401	<.0001
IKA*SUKUPUOLI	2	356.8512	<.0001

Analysis of Maximum Likelihood Estimates							
Parameter			DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept	1		1	3.3832	0.2469	187.8329	<.0001
Intercept	2		1	7.0431	0.2850	610.7024	<.0001
SUKUPUOLI	1		1	-2.2552	0.3712	36.9114	<.0001
SUKUPUOLI	2		0	0	.	.	.
VUOSI*SUKUPUOLI	2000	1	1	-0.4139	0.1205	11.7988	0.0006
VUOSI*SUKUPUOLI	2000	2	1	-0.3429	0.1069	10.2924	0.0013
VUOSI*SUKUPUOLI	2011	1	0	0	.	.	.
VUOSI*SUKUPUOLI	2011	2	0	0	.	.	.
IKA*SUKUPUOLI	1		1	-0.0403	0.00504	63.9957	<.0001
IKA*SUKUPUOLI	2		1	-0.0763	0.00432	310.9906	<.0001

SUDAAN Multilog / Cumulative Logit Model

R_SYSTBP2_123 (cum- logit), Independent Variables and Effects		Beta Coeff.	SE Beta	T-Test B=0	P-value T-Test B=0
R_SYSTBP2_123 (cum- logit)	Intercept 1	3.383	0.247	13.705	0.000
	Intercept 2	7.043	0.285	24.712	0.000
Sukuupuoli (1=M,2=N)	1	-2.255	0.371	-6.075	0.000
	2	0.000	0.000	.	.
Tutkimusvuosi, Sukuupuoli (1=M,2=N)	2000, 1	-0.414	0.121	-3.435	0.001
	2000, 2	-0.343	0.107	-3.208	0.001
	2011, 1	0.000	0.000	.	.
	2011, 2	0.000	0.000	.	.
Sukuupuoli (1=M,2=N), Ikä	1, 1	-0.040	0.005	-8.000	0.000
	2, 1	-0.076	0.004	-17.635	0.000

Contrast	Degrees of Freedom	Wald ChiSq	P-value Wald ChiSq
OVERALL MODEL	7.000	1570.971	0.000
MODEL MINUS INTERCEPT	5.000	366.641	0.000
SUKUPUOLI	.	.	.
VUOSI * SUKUPUOLI	2.000	22.240	0.000
IKA * SUKUPUOLI	2.000	356.851	0.000

Syst.vp (3-luok.)	Predicted Marginal #1	Predicted Marginal	Lower 95% Limit	Upper 95% Limit
1	Tutkimusvuosi, Sukupuoli (1=M,2=N) 2000, 1	0.216	0.189	0.245
	2000, 2	0.328	0.298	0.358
	2011, 1	0.289	0.251	0.330
	2011, 2	0.391	0.356	0.427
2	Tutkimusvuosi, Sukupuoli (1=M,2=N) 2000, 1	0.677	0.653	0.701
	2000, 2	0.574	0.549	0.598
	2011, 1	0.637	0.604	0.669
	2011, 2	0.534	0.504	0.564
3	Tutkimusvuosi, Sukupuoli (1=M,2=N) 2000, 1	0.107	0.088	0.130
	2000, 2	0.099	0.083	0.117
	2011, 1	0.074	0.060	0.091
	2011, 2	0.075	0.063	0.089

Direct Standardization: SUDAAN Descript + Stdvar & Stdwgt

```
TITLE2 "Direct Standardization: SUDAAN Descript + Stdvar & Stdwgt";
PROC DESCRIPT DATA=X.T0011;
  NEST  OSITE RYVAS;
  WEIGHT ALL_ANALYSIS_W;
  CLASS IKA_5LK VUOSI;
  STDVAR IKA_5LK;
  STDWGT 0.1 0.2 0.3 0.2 0.2; * sum=1;
  * or: STDWGT 10 20 30 20 20; * sum=100;
  * or: STDWGT 10000 20000 30000 30000 20000; * automatic scaling;
  VAR   R_SYSTBP2;
  TABLES VUOSI;
  PRINT / STYLE=NCHS;
RUN;
```

BEFORE STANDARDIZATION:

Variable	Tutkimusvuosi	Sample Size	Weighted Size	Mean	SE Mean	Lower 95% Limit Mean	Upper 95% Limit Mean
RR: syst,mittaus 2	Total	2432	2360.40	132.92	0.48	131.99	133.86
	2000	1452	1446.34	133.77	0.64	132.50	135.04
	2011	980	914.05	131.59	0.53	130.55	132.62

AFTER STANDARDIZATION:

Variable	Tutkimusvuosi	Sample Size	Weighted Size	Mean	SE Mean	Lower 95% Limit Mean	Upper 95% Limit Mean
RR: syst,mittaus 2	Total	2432	2360.40	137.24	0.52	136.22	138.25
	2000	1452	1446.34	139.74	0.77	138.23	141.24
	2011	980	914.05	134.09	0.55	133.01	135.17



TERVEYDEN JA
HYVINVOINNIN LAITOS
National Institute for Health and Welfare

Kiitää!